Formal Modeling and Verification of a Rate-Monotonic Scheduling Implementation with Real-Time Maude

***Abstract*—Rate-Monotonic Scheduling (RMS) is one of the most important real-time scheduling used in industry. There are a large number of results about RMS, especially on its schedulability. However, the theoretical results do not contain enough details to be used directly for an industrial RMS implementation. On the other hand, the correctness of such an implementation is of the crucial importance. In this paper, we analyze a realistic RMS implementation by using Real-Time Maude, a formal modeling language and analysis tool based on rewriting logic. Overhead and some details of the hardware are taken into account in the model. We validate the schedulability and the correctness of the implementation within key scenarios. The soundness and the completeness of our approach are substantiated.**

***Index Terms*—Real-time systems, scheduling, embedded systems, modeling, formal verification, rewriting logic.**

I. Introduction

PERIODIC task scheduling is one of the most important topics within the field of industrial real-time systems. A set of periodic tasks is said to be *schedulable* with respect to some scheduling algorithm if all jobs meet their deadlines. *Rate-Monotonic Scheduling* (*RMS*) is a *fixed* priority scheduling algorithm for preemptive hard real-time environments proposed by Liu and Layland [1], which assigns priorities to jobs according to the periods of the corresponding tasks: the smaller period, the higher priority. RMS is proven to be the *optimal* fixed priority scheduling algorithm [1], in the sense that any set of tasks, which is schedulable under *some* fixed priority scheduling algorithm, is also schedulable with respect to RMS. It is widely used in safety-critical real-time applications, such as vehicles and avionics, thanks to its optimality and easiness to implement.

Liu and Layland [1] gave a sufficient condition for the schedulability of a set of n tasks scheduled by RMS: $\Sigma_{i=1}^{n} C_i/T_i \leq n(2^{1/n} - 1)$, where C_i and T_i are the *computation (time) requirement* and the period of task τ_i, respectively. Two main directions on RMS have been explored since then. One is to relax the assumptions on the original RMS model, making it applicable on more systems. For instance, [2]–[5] allow aperiodic tasks in the scheduling, [6], [7] generalize RMS to be *deadline-monotonic*, [8] allows resource sharing among tasks, [9]–[12] extend RMS on multiprocessors, and [13]–[15] enhance fault-tolerance. The other direction is to generate better schedulability test conditions for the algorithm and its extensions [10], [12], [16]–[19]. The RMS algorithm is no doubt of practical importance.

It is even more crucial to ensure the reliability of an implementation instead of the algorithm, when RMS serves in a safety-critical system. When analyzing a realistic implementation, theoretical results may be no more applicable. For instance, even though the conditions derived from algorithm analysis are satisfied, schedulability can be broken by overhead in the system, or by the interrupt mask mechanism which may delay interrupt handling. On the other hand, correctness of the implementation with respect to the algorithm is difficult to be verified by the traditional methods such as testing and simulation due to their incompleteness. Extensive effort to apply formal methods, such as model checking and theorem proving, has been made to analyze safety-critical systems for the past few years [20]–[23]. However, as far as we know, few [24], [25] attempt to analyze the RMS algorithm, while no work for implementations of RMS is found.

In this paper, we use *Real-Time Maude*, a *rewriting*-based formal modeling language and analysis tool for real-time systems, to model a realistic implementation of RMS, which serves as a simplified operating system within an avionic control system, and then verify several desired properties. Based on a realistic implementation, our model extends the standard RMS model proposed in [1], by considering overhead and other details of the hardware platform.

The rest of this paper is organized as follows. Section II gives some background of both the RMS algorithm and Real-Time Maude. Section III presents the RMS implementation that we model and analyze. Section IV introduces how we model the RMS implementation using Real-Time Maude. Then Section V explains how to verify the desired properties and to evaluate the results. Related work is discussed in Section VI. We conclude the paper in Section VII.

II. Background

A. Rate-Monotonic Scheduling Algorithm

A task set consists of *only* n periodic tasks $\tau_1, \ldots, \tau_n$. Each task τ_i has a period T_i and a computation requirement C_i. First jobs of all tasks are assumed to be initiated at time 0 simultaneously. Deadlines consist of runnability constraints only: the deadline of a job corresponding to τ_i is the initiation time of the next job corresponding to τ_i. The RMS algorithm chooses the labeling such that $T_1 \leq T_2 \leq \ldots \leq T_n$. Consequently, τ_i receives priority i, assuming smaller numbers have higher priorities. The following assumptions are made:

(A1) Jobs corresponding to task τ_i are initiated exactly at times kT_i with integers $k \geq 0$.

(A2) Computation requirement C_i for each task τ_i is constant and does not vary with time.

(A3) Tasks are independent, such that they are ready to run at their initiation times and can be preempted instantly (ignoring all blocking).

(A4) All overhead, such as task switching time, is ignored.

A simple example showing this algorithm is depicted in Figure 2a. However in this paper, we consider an implementation instead of the RMS algorithm itself, thus the model would be more complicated than this standard, ideal setting. Assumption (A1) will be modified because of the interrupt mask mechanism, while (A4) is relaxed to obtain a more realistic analysis model.

B. Real-Time Maude

Real-Time Maude [26] is an extension of *Maude* [27] which is a language and tool based on *rewriting logic* [28]. It supports formal specification and analysis of real-time systems.

1) Specification: Real-Time Maude models systems as *modules*. A module specifies a *real-time rewrite theory* $\mathcal{R} = (\Sigma, E \cup A, IR, TR)$, where:

- Σ is an algebraic *signature*, that is, a set of declarations of *sorts*, *subsorts* and *function symbols*. The function symbols are allowed to be mixfix.
- $(\Sigma, E \cup A)$ is a *membership equational logic theory*, with E a set of conditional *equations* and *memberships* on Σ, and A a set of equational axioms such as associativity, commutativity and identity. $(\Sigma, E \cup A)$ models the system's "static" states as an algebraic data type, and is equipped with a built-in specification of a sort `Time`.
- IR is a set of *labeled conditional rewrite rules* specifying the system's local transitions. Each rule has the form $[l] \ : \ s \rightarrow s'$ **if** $\bigwedge_{j=1}^{n} cond_j$, where each $cond_j$ is an equality $u_j = v_j$ and l is a *label*. Such a rule specifies an *instantaneous transition*, without consuming time, from an instance of s to the corresponding instance of s', *provided* the conditions hold.
- TR is a set of (*labeled*) *tick rules* of the form $[l] \ : \ \{s\} \rightarrow \{s'\}$ **in time** t **if** $cond$ that specify *timed transitions*, which advances time in the *entire* state s by t time units. IR and TR together model the "dynamic" behaviors of the system.

In rewriting logic, rewrite rules are applied non-deterministically, that is, when several rules can be applied on a given s, any of them may be chosen. Hence non-deterministic behaviors can be modeled naturally in Real-Time Maude. Real-Time Maude also supports specifications in *object-oriented* style. A class declaration `class` C `|` att_1`:`s_1`,` $\ldots$`,` att_n`:`s_n defines a class C with attributes att_1 to att_n of sorts s_1 to s_n, respectively. An *object* of class C is represented as a term `<` O`:`C `|` att_1`:`val_1`,` $\ldots$`,` att_n`:`val_n `>` of sort `Object`, where O, of sort `Oid`, is the object's *identifier*, and val_i is the value of the attribute att_i with $i \in [1, n]$. A *subclass* inherits all the attributes and rules of its superclasses.

2) Formal Analysis: Real-Time Maude provides many useful commands and tools to analyze a given model. For example, `rewrite` allows to execute the model, symbolically; given an initial state, `search` is used to search reachable states satisfying the desired properties; the Maude's *Inductive Theorem Prover* (ITP) can be applied to interactively prove properties written in *membership equational logic*.

In this paper, we only consider Real-Time Maude's *Linear Temporal Logic (LTL) model checker*, which analyzes whether *each* behavior satisfies a temporal logic formula. *State propositions* are defined as terms of sort `Prop`. Their semantics is defined by conditional equations of the form `ceq` $statePattern$ `|=` $prop$ `=` b `if` $cond$, with b a term of sort `Bool`, stating that $prop$ evaluates to b in states which are instances of $statePattern$ provided the condition $cond$ holds. These equations together define $prop$ to hold in all states s that make s `|=` $prop$ evaluate to `true`. A temporal logic *formula* is constructed by state propositions and temporal logic operators such as `True`, `False`, `~`(negation), `/\`, `\/`, `->`(implication), `[]`("always") and `U`("until"). Real-Time Maude supports both *timed* and *untimed LTL model checking*. The untimed model checking command

```
(mc s |=u Φ .)
```

checks whether the temporal logic formula Φ holds in all behaviors starting from the initial state s, *with no time limit*.

III. The Implementation of RMS

The implementation is in an industrial avionic control system. Interrupts would be triggered by, and only by, the clock every T, which we call *interrupt cycle*. When an interrupt request occurs, if the system is interruptible, i.e. the interrupt mask bit is cleared, the handler function $schedule()$ will be invoked; otherwise $schedule()$ will be pending until the interrupt mask bit becomes cleared. The implementation is shown as $schedule()$ in Figure 1, where $taskList$ is the list of periodic tasks to be scheduled. We assume that the list is in descending order of priority, and both variables $taskList$ and $timer$ are global. In this implementation, there is only one kind of interrupt, the period T_i of each task is a multiple of T, and the tasks are independent, meeting assumption (A3).

In Figure 1, the handler function $schedule()$ first updates status of all tasks in $taskList$ via function $updateStatus()$. This updating actually initiates tasks that should be scheduled in the current interrupt cycle. Then $schedule()$ traverses the list to execute the ready tasks one by one, or to do a return when encountering an interrupted[1] task. As for the function $updateStatus()$, it updates each task in two steps: firstly, if the task is running, it becomes interrupted; secondly for the task at its initiation time, if its previous job is complete, it would be set ready, otherwise it misses its deadline, producing an error. Notice that $schedule()$ is invoked only when the interrupt request is handled, not when the interrupt is disabled (the interrupt mask bit is set). Due to the interrupt mask bit, its execution cannot be interrupted when it is updating status of tasks or searching the next task to execute, however, it can be interrupted while executing some task (Line 12). This allows the execution of $schedule()$ to be nested.

[1]Note that the status *INTERRUPT* indicates the task is interrupted for the moment, or was interrupted before but its execution is not complete yet.

```
1: function schedule()
2:     int_off(); ▷ to disable interrupts
3:     updateStatus(taskList);
4:     timer = timer + 1;
5:     p = taskList;
6:     while p do
7:         if p → status == INTERRUPT then
8:             return ;
9:         else if p → status == READY then
10:            p → status = RUNNING;
11:            int_on(); ▷ to enable interrupts
12:            p → function(); ▷ to execute the task
13:            int_off();
14:            p → status = DORMANT;
15:        end if
16:        p = p → next;
17:    end while
18: end function
19: function updateStatus(p)
20:    while p do
21:        if p → status == RUNNING then
22:            p → status = INTERRUPT;
23:        end if
24:        if timer % (p → period) == 0 then ▷ the task should
       be initiated
25:            if p → status == DORMANT then ▷ the previous
       job finishes
26:                p → status = READY;
27:            else ▷ the status is READY or INTERRUPT
28:                reportTaskError(p); ▷ task misses its deadline
29:            end if
30:        end if
31:        p = p → next;
32:    end while
33: end function
```

Fig. 1. The C-Like Pseudo-code of $schedule()$

For simplicity, in the rest of this paper, we use *scheduling* to refer to the stage, from the moment when a pending interrupt request is detected, to the moment when the first should-be-run periodic task starts executing, i.e. Line 8 or 12 in Figure 1. Therefore, *scheduling time* mainly consists of three parts: (i) the time for switching context from the running task, possibly none, to $schedule()$ when an interrupt request is handled, (ii) the time spent by $schedule()$ searching and setting the *first* should-be-run periodic task (Lines 2-11 in Figure 1), and (iii) the time for switching context from $schedule()$ to that task. *Switching* refers to the stage, from the moment when a periodic task completes its execution, to the moment when the next should-be-run periodic task starts executing. *Switching time* thus also consists of three parts: (i) the time for switching context from the complete task back to $schedule()$, (ii) the time spent by $schedule()$ searching and setting the *next* should-be-run periodic task, and (iii) the time for switching context from $schedule()$ to that task.

IV. Formal Modeling of the Implementation

Considering some technical details of the platform, such as interrupt mask mechanism, we model the implementation under the following assumptions:



Fig. 2. A set of 2 periodic tasks scheduled under RMS algorithm and the implementation, respectively. $T = T_1 = 10$, $T_2 = 20$, $C_1 = 3$, $C_2 = 2$. In Figure 2b, we assume both scheduling time and switching time are 2.

(A1') Jobs corresponding to task τ_i are initiated at the beginning of scheduling that handles the requests triggered at times kT_i with integers $k \geq 0$.

(A2) Computation requirement C_i for each task τ_i is constant and does not vary with time.

(A3) Tasks are independent, such that they are ready to run at their initiation times and can be preempted instantly.

(A4') Scheduling time and switching time are considered, while other overhead is ignored.

These assumptions make our model different from the standard one. For instance, with assumption (A1'), an interrupt request that occurs during switching will be pending, such that jobs of task τ_i cannot initiate at kT_i. They should wait until switching finishes and the interrupt mask bit is cleared, which is different from (A1). Note that (A3) says that jobs are ready to run at their initiation times, however, no one can start running at exactly its initiation time, because scheduling takes time. Under these assumptions, an example showing the execution of tasks scheduled by the implementation is depicted in Figure 2b. Note that the second job instance of τ_1 is initiated at time 11 instead of 10, because switching is performed and the interrupt mask bit is set at time 10.

In this section, we introduce first how we model the states–the static aspect–of the system using algebraic data types, then how we specify the essential behaviors–the dynamic aspect–using rewrite rules. In particular, modeling of *instantaneous behaviors* would be explained in Sections IV-C, IV-D and IV-E, followed by *timed behaviors* in Section IV-F.

A. Basic Data Types

In our model, tasks are identified by their indexes of sort `Nat` in the $taskList$. We define a sort `MaybeNat` wrapping `Nats` to refer to some task, with *constructor* `some` followed by a `Nat` n indicating the task indexed n, and `none` for no task:

```
op none : -> MaybeNat [ctor] .
```

```
op some_ : Nat -> MaybeNat [ctor] .
```

A sort `Stack` is introduced to model the stack of the system, storing the tasks that are being interrupted, and equipped with operations `push`, `pop` and `peek` on it.

```
op bottom : -> Stack [ctor] .
op _#_ : Nat Stack -> Stack [ctor] .
```

We also need a sort `Counter` to record the execution of tasks. We call *execution time* the time how long a task has been executing for. A `Counter` records the execution time and the computation requirement of a task.

```
op [_/_] : Time Time -> Counter [ctor] .
```

The global variable $timer$ is reset when it reaches an upper bound while increasing, which is not shown in detail in Figure 1 but is reasonable. The upper bound is the least common multiple of the periods of all tasks. A sort `Timer` is defined:

```
op [_/_] : Nat NzNat -> Timer [ctor] .
```

B. Modeling the System States

The system can be considered as consisting of several parts: the tasks which are scheduled, the scheduler itself, the hardware including registers and stacks, and the interrupt source. The scheduler, i.e. the function $schedule()$, can be described by a single variable $timer$. We present the models of the other parts one by one.

1) Tasks: Each task is abstracted from its functionality as a `Counter`. Overhead for scheduling and switching is considered in our model. They are treated as two system tasks. Every task is modeled as an object instance of some subclass of the base class `Task`:

```
class Task | cnt : Counter .
op error : -> Object [ctor] .
```

where `error` is an object indicating some task that misses its deadline.

A periodic task, which needs to be scheduled, is an object instance of the subclass `PTask` of `Task` with additional attributes `priority`, `period` and `status`:

```
class PTask | priority : Nat, period : NzNat,
              status : Status .
subclass PTask < Task .
```

where `Status` is a sort with four constant constructors, same as in the implementation:

```
ops RUNNING INTERRUPT READY DORMANT
      : -> Status [ctor] .
```

The list of periodic tasks, the variable $taskList$ in the implementation, is modeled as an instance of the sort `TaskList`, which is a list of `PTasks`[2]:

```
op null : -> TaskList [ctor] .
op _::_ : Object TaskList
            ~> TaskList [ctor] .
mb (< O:Oid : PTask |> :: L:TaskList)
     : TaskList .
mb (error :: L:TaskList) : TaskList .
```

Periodic tasks are identified by their indexes in the list.

[2]Following the Maude convention, variables would be written in capital letters.

On the other hand, a system task is an object instance of the subclass `SysTask` of `Task` with no extra attributes:

```
class SysTask | .
subclass SysTask < Task .
```

Different from periodic tasks, system tasks are organized in a multiset of sort `SysTasks`, and identified by their `Oids`.

2) Hardware: Our model considers two parts of hardware related to the interrupt handling mechanism: the registers and the stack.

The set of registers is modeled as an object instance of the class `Regs`, with attributes `pc` denoting the program counter, `mask` for the interrupt mask bit and `ir` for the interrupt request bit, respectively.

```
class Regs | pc : TaskID,
             mask : Bool, ir : Bool .
```

where the sort `TaskID` is a supersort of `MaybeNat` and `Oid`, referring to some task.

```
subsorts MaybeNat Oid < TaskID .
```

Some operations, such as `getPc` and `setMask`, are defined on the class `Regs`.

Then the hardware is described by the sort `Hardware`:

```
op [_;_] : Object Stack ~> Hardware [ctor] .
mb ([ < O:Oid : Regs |> ; S:Stack ])
     : Hardware .
```

3) Interrupt Source: The interrupt source is modeled as an object of class `IntSrc`, with attributes `cycle` denoting the interrupt cycle T, and `val` for the value which will decrease from T to 0 while time advances:

```
class IntSrc | val : Time, cycle : Time .
```

4) System: The entire system in our model is a composition of the parts introduced above, of a sort `System`[3]:

```
op _____ : TaskList Timer
           SysTasks Hardware Object
             ~> System [ctor] .
mb (L T STS HW < O : IntSrc |>) : System .
```

C. Interrupt Requests

Interrupt requests are performed by the source exactly every cycle T, when the attribute `val` decreases to `zero`. The requesting is an instantaneous action, thus is modeled by the following instantaneous conditional rule applied on `System`:

```
crl [interrupt-request] :
  (L T STS HW ISRC)
  => (L T STS (HW).intReq reset(ISRC))
  if (ISRC).timeout .
```

where the function `_.timeout` examines whether the attribute `val` equals `zero`, and `_.intReq` sets the `ir` bit indicating there exists an interrupt request to be handled:

```
op _.intReq : Hardware -> Hardware .
eq [ REGS ; S ].intReq
     = [ (REGS).setIr ; S ] .
```

Then the request will wait to be handled, which is explained in Section IV-E.

[3]Some variable declarations are not shown for simplicity.

D. Task Initiation

Periodic tasks are initiated sequentially by the function $updateStatus()$ in Figure 1, which is treated as an instantaneous action in our model. It is modeled by the recursive function as below:

```
op updateStatus_with_ : TaskList Timer
                        -> TaskList .
eq updateStatus null with TIMER = null .
eq updateStatus (TASK :: L) with TIMER
   = (update TASK with TIMER)
     :: (updateStatus L with TIMER) .
```

with `TIMER` the current value of the global variable $timer$, and function `update_with_` updating the status of individual task (Lines 21-30 in Figure 1):

```
op update_with_ : Object Timer ~> Object .
ceq update < O : PTask | period : T,
                         status : ST >
      with TIMER
    = if ST == DORMANT
      then < O : PTask | status : READY >
      else error fi
    if TIMER rem T == 0 .
eq update < O : PTask | status : ST >
     with TIMER
   = if ST == RUNNING
     then < O : PTask | status : INTERRUPT >
     else < O : PTask |> fi [owise] .
```

Given a task, if `TIMER`($timer$) can be divided by its period, this task should be initiated. In the case where the task should be initiated, it is set `READY` if its `status` is `DORMANT`; otherwise that means the previous job of this task is not complete, hence it misses its deadline, producing an `error`. In the other case where the task should not be initiated, its `status` changes only if it is `RUNNING`.

We can see that `updateStatus_with_` behaves the same as $updateStatus()$.

E. Interrupt Handling and Task Scheduling

When an interrupt request occurs, it may not be detected immediately by the system. It requires the bit `mask` to be cleared. Once the request is detected, it is handled in two steps: the interrupt handling mechanism of the hardware (such as clearing `ir`, pushing context into stack and so on), and to invoke the function $schedule()$. This behavior is modeled by the following instantaneous rewrite rule:

```
crl [interrupt-handle] :
  SYSTEM
  => ((SYSTEM).interrupt).startScheduling
  if (SYSTEM).existInt .
```

where `_.existInt` checks whether `mask` is cleared *and* `ir` is set. The function `_.interrupt` models the interrupt handling mechanism performed by the hardware and does four things: (i) clearing the bit `ir`, which means the request has been handled; (ii) pushing the current `pc` value into the stack, storing the interrupted context; (iii) assigning `scheduling` of sort `Oid` to `pc`, which indicates that the system is scheduling; and (iv) setting the bit `mask`, to mask coming interrupt requests.

Unlike periodic tasks, even though the scheduling stage is modeled by a `Counter`, its functionality is too important to be abandoned. We divide the behaviors of scheduling into three parts. The first part contains its timed behaviors. This part is modeled by regarding scheduling as a system task of sort `SysTask`. Modeling timed behaviors of tasks is explained in Section IV-F. The other two parts together define its functionality. The second part corresponds to Lines 3-4 in Figure 1. It updates the status of $taskList$ and increases $timer$ by 1. This part is modeled by function `_.startScheduling` which applies instantaneously at the beginning of scheduling, as shown in rule `interrupt-handle`:

```
op _.startScheduling : System -> System .
eq (L T STS HW ISRC).startScheduling
   = ((updateStatus L with T)
      inc(T) STS HW ISRC) .
```

The third part corresponds to Lines 6-11, searching the first should-be-run periodic task and setting it to execute. It is modeled by function `_.finishScheduling`, which applies instantaneously at the end of scheduling:

```
op _.finishScheduling : System -> System .
eq (L T STS HW ISRC).finishScheduling
   = (L T (finish scheduling in STS)
      HW ISRC).run1stTask .
```

where `finish_in_` resets the counter of task `scheduling`, and `_.run1stTask` models Lines 6-11 in Figure 1, searching the task with highest priority that has status `INTERRUPT` or `READY` then performing an *interrupt return* or executing it, respectively.

When the execution time of the system task `scheduling` reaches its computation requirement, the scheduling stage may finish. We model this instantaneous action with the following rule:

```
crl [scheduling-finish] :
  (L T STS HW ISRC)
  => (SYSTEM).finishScheduling
  if SYSTEM := (L T STS HW ISRC)
     /\ (SYSTEM).running == scheduling
     /\ scheduling isComplete?in STS .
```

where function `_.running` returns the current `pc` value of the system, and `_isComplete?in_` checks whether the execution time of the task reaches its computation requirement.

Similar to scheduling, the switching stage is divided into timed behaviors of `switching` and its functionality. `switching` starts when the execution time of the running periodic task reaches its computation requirement, and finishes when its own execution time does so. Two similar instantaneous rules are defined to model the functionality of switching:

```
crl [task-finish] :
  (L T STS HW ISRC)
  => (SYSTEM).startSwitching
  if SYSTEM := (L T STS HW ISRC)
     /\ some N := (SYSTEM).running
     /\ some N isComplete?in L .
crl [switching-finish] :
  (L T STS HW ISRC)
  => (SYSTEM).finishSwitching
  if SYSTEM := (L T STS HW ISRC)
     /\ (SYSTEM).running == switching
     /\ switching isComplete?in STS .
```

F. Timed Behaviors of the System

Timed behaviors of the system consist of two parts: the execution of tasks and the execution of the interrupt source. Both are modeled together by the following single "standard" tick rule [29]:

```
crl [tick]:
  {SYSTEM} => {delta(SYSTEM, R)} in time R
  if R le mte(SYSTEM) [nonexec] .
```

where `delta` defines the effects of time elapse on the system, and `mte` denotes the *m*aximum amount of *t*ime allowed to *e*lapse from the current state until an instantaneous transition *must* be performed. In fact, the core to model timed behaviors is to define functions `delta` and `mte`. Notice that the variable `R` is *continuous* with respect to the specific time domain[4] that we choose to instantiate our model on, which is different from timed automata that discretize dense time by defining "clock region".

Time affects the system by advancing both the running task whose ID is loaded at `pc` and the interrupt source simultaneously. While time elapses, `cnt` of the former increases and `val` of the latter decreases, respectively:

```
ceq delta((L T STS HW ISRC), R)
    = ((deltaTask(ID, L, R)
       T STS HW (deltaIS(ISRC, R)))
    if ID := (HW).getPc /\ ID :: MaybeNat .
```

We omit details for the case where `ID` is of sort `Oid` due to similarity. In that case, `deltaTask` applies on `STS` instead of `L`.

`mte`, the maximum amount of time allowed to elapse, depends on when the next instantaneous action must perform. Therefore, it is decided by three arguments: the remaining time to complete the running task, the remaining time to request the next interrupt, and whether or not there exists an interrupt request detected for the moment:

```
ceq mte(L T STS HW ISRC)
    = minimum( mteTask(ID, L),
               minimum( mteIS(ISRC),
                        mteIr(HW)))
    if ID := (HW).getPc /\ ID :: MaybeNat .
```

where `mteIr` returns `zero` if there exists an interrupt request detected in the system, or `INF` which represents *infinity* otherwise. Again we do not show the case where `ID` is of sort `Oid`, which is very similar.

V. Formal Verification

In this section, we analyze our model of the RMS implementation within different realistic scenarios. Notice that from any (reasonable) given initial state, the number of reachable states is finite, but may be unknown, thanks to the upper bound given to $timer$, which provides the potential for applying the untimed model checker.

A. Properties

We consider two properties in this paper: schedulability and correctness. By schedulability, we examine whether a given task set is schedulable by the implementation. By correctness, we verify whether the implementation schedules periodic tasks exactly with respect to their priorities and periods.

To verify the schedulability of a given set of periodic tasks, we define an atomic proposition `taskTimeout` to hold if there exists an `error` appearing in $taskList$ of the current state, that is, some task misses its deadline:

```
op taskTimeout : -> Prop [ctor] .
eq {L T STS HW ISRC} |= taskTimeout
   = containError(L) .
```

where `containError` returns `true` if there is an object `error` existing in `L`. Then our desired property–schedulability–can be formalized as the temporal logic formula: `[](~taskTimeout)`. As the property is not *clocked*, given an initial state `init`, the following untimed model checking command returns `true` if the schedulability property holds with no time limit; otherwise a trace showing a counterexample is provided:

```
(mc init |=u [](~taskTimeout) .)
```

Another objective is to verify the correctness of the implementation. The atomic proposition `correct` is hence defined to hold if the running periodic task is the one requested to be executed with the highest priority:

```
op correct : -> Prop [ctor] .
ceq {L T STS HW ISRC} |= correct
    = if ID :: MaybeNat then shouldRun(ID, L)
      else true fi
    if ID := (HW).getPc .
```

where `shouldRun(ID, L)` returns `true` if the task identified by `ID`, probably `none`, is the one possessing the highest priority among those whose status is `READY`, `RUNNING` or `INTERRUPT`. Note that in this paper, we do not care the behaviors after some task misses its deadline. Therefore, the correctness property is formalized by the temporal logic formula: `([]correct) \/ (correct U taskTimeout)`, and can be verified by the following untimed model checking command provided an initial state `init`:

```
(mc init |=u ([]correct)
             \/ (correct U taskTimeout) .)
```

B. Scenarios

We use the following setting for our verification, which is from the statistics provided by our industrial partner:

- The interrupt cycle T is $5ms$.
- The scheduling time is $38\mu s$ and the switching time is $20\mu s$.
- The initial state is with empty stack, empty `pc`, cleared `mask` and cleared `ir`.

We have analyzed our model/implementation in ten different scenarios, including both realistic ones provided by our industrial partner and experimental ones designed by ourselves, four of them are described below:

- Scenario (i) with two tasks τ_1 and τ_2: $T_1 = 5ms$, $C_1 = 3ms$, $T_2 = 25ms$ and $C_2 = 7ms$.
- Scenario (ii) with two tasks τ_1 and τ_2: $T_1 = 5ms$, $C_1 = 2ms$, $T_2 = 25ms$ and $C_2 = 2.3ms$.
- Scenario (iii) with three tasks τ_1, τ_2 and τ_3: $T_1 = 5ms$, $C_1 = 2.7ms$, $T_2 = 10ms$, $C_2 = 2ms$, $T_3 = 25ms$ and $C_3 = 3ms$.
- Scenario (iv) with three tasks τ_1, τ_2 and τ_3: $T_1 = 5ms$, $C_1 = 2.5ms$, $T_2 = 10ms$, $C_2 = 1.5ms$, $T_3 = 15ms$ and $C_3 = 4.5ms$.

Instantiating our model on dense time domain and choosing the *maximal time sampling strategy*, the results of the

[4]Real-Time Maude contains built-in modules to define the time domain to be natural numbers and rational numbers, specifying *discrete* time domains and *dense* time domains, respectively.



Fig. 3. A counterexample of schedulability in Scenario (iv). The first job instance of τ_3 misses its deadline at time $15ms$, which is the initiation time for the second job instance of τ_3.

model checking show that the correctness property holds in all scenarios. As to the schedulability property, it holds in Scenarios (i-iii), but fails in Scenario (iv). One counterexample of the schedulability for Scenario (iv), returned by the model checking command, is pictured in Figure 3, where the third task τ_3 misses its deadline at the time $15ms$.

C. Evaluation

We now show in this section that our results are both sound and complete.

An analysis method is called *sound* if any counterexample found using such a method is a real counterexample of the question, and *complete* if the fact that no counterexample can be found using such a method implies no counterexample exists for the question in analysis. The soundness of our results is trivial to check, simply by examining the counterexamples found. For instance, the counterexample shown in Figure 3 is a real counterexample, implying that the result for schedulability of Scenario (iv) is sound. But this is not the case for completeness, since we choose instantiating our model on dense time domain to make it more real but giving rise to an infinite state space which is unfeasible to exhaust.

In general, completeness of untimed model checking cannot be achieved for any systems, any time sampling strategies and any properties. However, Ölveczky and Meseguer proved the completeness of untimed temporal logic model checking, under the maximal time sampling strategy, for a large class of real-time systems [29]:

Theorem 1 ([29]): Given a *time-robust* real-time rewrite theory $\mathcal{R}$, a set AP of *tick-stabilizing* atomic propositions, an LTL formula Φ (excluding the *next* operator $\bigcirc$) whose atomic propositions are contained in AP. The untimed temporal logic model checking verifying Φ is *complete* under the maximal time sampling strategy.

Therefore, we achieve the following theorem, showing that the results in Section V-B are complete.

Theorem 2: Our approach using untimed model checking to verify the schedulability and the correctness of our model is complete.

Proof: By showing that our model is time-robust and that the two defined atomic propositions–`taskTimeout` and `correct`–are tick-stabilizing, then using Theorem 1. ■

VI. Related Work

In this section we discuss our results with related work in three directions.

Considering schedulability test, Liu and Layland [1] gave the famous sufficient condition that a set of periodic tasks is schedulable with respect to RMS if $\Sigma_{i=1}^{n} C_i/T_i \leq n(2^{1/n}-1)$ holds. Then a more sufficient condition, known as *Hyperbolic Bound*, which has the same complexity as Liu and Layland's, was proposed in [18]. On the other hand, necessary and sufficient conditions for schedulability were derived independently in [3] and [7], requiring more sophisticated analysis on the task set. Nevertheless, all these results take no overhead into account, being not as realistic as ours. Katcher et al. did consider overhead in their schedulability analysis, under several models based on different kinds of popular implementations [30]. However, our target implementation is not in their scope. Furthermore, compared with those theoretical analyses, our approach based on formal modeling and verification has a great advantage that if our schedulability test answers "no", it returns at the same time a real counterexample, which is able to guide our engineer to adjust the design.

[24] and [25] also made use of model checking to analyze the RMS algorithm along the same line but with different languages and tools. They considered only the ideal setting as in [1], instead of an implementation which contains much more complex details. Our model of the implementation easily degenerates to a model of the RMS algorithm, if we let the times for scheduling and switching be zero.

Finally, Maude and Real-Time Maude have been successfully applied on large numbers of applications [28], especially on communication protocols, real-time and cyber-physical systems. But few results are achieved on scheduling problems. RMS is investigated using Real-Time Maude for the first time.

VII. Conclusion

We have formally modeled a realistic implementation of RMS using Real-Time Maude, a modeling language based on rewriting logic. By taking into account the overhead of scheduling and switching, and by modeling some mechanism of the hardware, our model contains sufficient details to be analyzed for the behaviors of real systems. Two important properties–schedulability and correctness–are verified by model checking on our model, within several key scenarios. We demonstrate the soundness and completeness of our results.

References

[1] C. L. Liu and J. W. Layland, "Scheduling algorithms for multiprogramming in a hard-real-time environment," *J. ACM*, vol. 20, no. 1, pp. 46–61, 1973.

[2] J. P. Lehoczky, L. Sha, and J. K. Strosnider, "Enhanced aperiodic responsiveness in hard real-time environments," in *Proceedings of the 8th IEEE Real-Time Systems Symposium (RTSS '87), December 1-3, 1987, San Jose, Califorinia, USA*. IEEE Computer Society, 1987, pp. 261–270.

[3] B. Sprunt, L. Sha, and J. P. Lehoczky, "Aperiodic task scheduling for hard real-time systems," *Real-Time Systems*, vol. 1, no. 1, pp. 27–60, 1989.

[4] J. P. Lehoczky and S. Ramos-Thuel, "An optimal algorithm for scheduling soft-aperiodic tasks in fixed-priority preemptive systems," in *Proceedings of the Real-Time Systems Symposium - 1992, Phoenix, Arizona, USA, December 1992*. IEEE Computer Society, 1992, pp. 110–123.

[5] J. K. Strosnider, J. P. Lehoczky, and L. Sha, "The deferrable server algorithm for enhanced aperiodic responsiveness in hard real-time environments," *IEEE Trans. Computers*, vol. 44, no. 1, pp. 73–91, 1995.

[6] J. Y. Leung and J. Whitehead, "On the complexity of fixed-priority scheduling of periodic, real-time tasks," *Perform. Eval.*, vol. 2, no. 4, pp. 237–250, 1982.

[7] N. Audsley, A. Burns, and A. Wellings, "Deadline monotonic scheduling theory and application," *Control Engineering Practice*, vol. 1, no. 1, pp. 71–78, 1993.

[8] L. Sha, R. Rajkumar, and J. P. Lehoczky, "Priority inheritance protocols: An approach to real-time synchronization," *IEEE Trans. Computers*, vol. 39, no. 9, pp. 1175–1185, 1990.

[9] S. K. Dhall and C. Liu, "On a real-time scheduling problem," *Operations Research*, vol. 26, no. 1, pp. 127–140, 1978.

[10] J. M. López, M. García, J. L. Díaz, and D. F. García, "Utilization bounds for multiprocessor rate-monotonic scheduling," *Real-Time Systems*, vol. 24, no. 1, pp. 5–28, 2003.

[11] J. M. López, J. L. Díaz, and D. F. García, "Minimum and maximum utilization bounds for multiprocessor rate monotonic scheduling," *IEEE Trans. Parallel Distrib. Syst.*, vol. 15, no. 7, pp. 642–653, 2004.

[12] S. K. Baruah and J. Goossens, "Rate-monotonic scheduling on uniform multiprocessors," *IEEE Trans. Computers*, vol. 52, no. 7, pp. 966–970, 2003.

[13] Y. Oh and S. H. Son, "Enhancing fault-tolerance in rate-monotonic scheduling," *Real-Time Systems*, vol. 7, no. 3, pp. 315–329, 1994.

[14] S. Ghosh, R. G. Melhem, D. Mossé, and J. S. Sarma, "Fault-tolerant rate-monotonic scheduling," *Real-Time Systems*, vol. 15, no. 2, pp. 149–181, 1998.

[15] A. A. Bertossi, L. V. Mancini, and F. Rossini, "Fault-tolerant rate-monotonic first-fit scheduling in hard-real-time systems," *IEEE Trans. Parallel Distrib. Syst.*, vol. 10, no. 9, pp. 934–945, 1999.

[16] J. P. Lehoczky, L. Sha, and Y. Ding, "The rate monotonic scheduling algorithm: Exact characterization and average case behavior," in *Proceedings of the Real-Time Systems Symposium - 1989, Santa Monica, California, USA, December 1989*. IEEE Computer Society, 1989, pp. 166–171.

[17] T. Kuo and A. K. Mok, "Load adjustment in adaptive real-time systems," in *Proceedings of the Real-Time Systems Symposium - 1991, San Antonio, Texas, USA, December 1991*. IEEE Computer Society, 1991, pp. 160–170.

[18] E. Bini, G. C. Buttazzo, and G. M. Buttazzo, "Rate monotonic analysis: The hyperbolic bound," *IEEE Trans. Computers*, vol. 52, no. 7, pp. 933–942, 2003.

[19] M. K. Gardner, "Probabilistic analysis and scheduling of critical soft real-time systems," 1999, Ph.D. thesis.

[20] F. Miyawaki, K. Masamune, S. Suzuki, K. Yoshimitsu, and J. Vain, "Scrub nurse robot system-intraoperative motion analysis of a scrub nurse and timed-automata-based model for surgery," *IEEE Transactions on Industrial Electronics*, vol. 52, no. 5, pp. 1227–1235, 2005.

[21] J. Meseguer and G. Rosu, "The rewriting logic semantics project: A progress report," *Inf. Comput.*, vol. 231, pp. 38–69, 2013.

[22] X. Leroy, "Formal verification of a realistic compiler," *Commun. ACM*, vol. 52, no. 7, pp. 107–115, 2009.

[23] G. Klein, K. Elphinstone, G. Heiser, J. Andronick, D. Cock, P. Derrin, D. Elkaduwe, K. Engelhardt, R. Kolanski, M. Norrish, T. Sewell, H. Tuch, and S. Winwood, "sel4: formal verification of an OS kernel," in *Proceedings of the 22nd ACM Symposium on Operating Systems Principles 2009, SOSP 2009, Big Sky, Montana, USA, October 11-14, 2009*, J. N. Matthews and T. E. Anderson, Eds. ACM, 2009, pp. 207–220.

[24] J. Cui, Z. Duan, and C. Tian, "Model checking rate-monotonic scheduler with TMSVL," in *2014 19th International Conference on Engineering of Complex Computer Systems, Tianjin, China, August 4-7, 2014*. IEEE Computer Society, 2014, pp. 202–205.

[25] C. Tian and Z. Duan, "Model checking rate monotonic scheduling algorithm based on propositional projection temporal logic," *Journal of Software*, vol. 22, no. 2, pp. 211–221, 2011, in Chinese.

[26] P. C. Ölveczky and J. Meseguer, "Semantics and pragmatics of real-time maude," *Higher-Order and Symbolic Computation*, vol. 20, no. 1-2, pp. 161–196, 2007.

[27] M. Clavel, F. Durán, S. Eker, P. Lincoln, N. Martí-Oliet, J. Meseguer, and J. F. Quesada, "Maude: specification and programming in rewriting logic," *Theor. Comput. Sci.*, vol. 285, no. 2, pp. 187–243, 2002.

[28] J. Meseguer, "Twenty years of rewriting logic," *J. Log. Algebr. Program.*, vol. 81, no. 7-8, pp. 721–781, 2012.

[29] P. C. Ölveczky and J. Meseguer, "Abstraction and completeness for real-time maude," *Electr. Notes Theor. Comput. Sci.*, vol. 176, no. 4, pp. 5–27, 2007.

[30] D. I. Katcher, H. Arakawa, and J. K. Strosnider, "Engineering and analysis of fixed priority schedulers," *IEEE Trans. Software Eng.*, vol. 19, no. 9, pp. 920–934, 1993.